/*M///////////////////////////////////////////////////////////////////////////////////////
//
//  IMPORTANT: READ BEFORE DOWNLOADING, COPYING, INSTALLING OR USING.
//
//  By downloading, copying, installing or using the software you agree to this license.
//  If you do not agree to this license, do not download, install,
//  copy or use the software.
//
//
//                           License Agreement
//                For Open Source Computer Vision Library
//
// Copyright (C) 2010-2012, Institute Of Software Chinese Academy Of Science, all rights reserved.
// Copyright (C) 2010-2012, Advanced Micro Devices, Inc., all rights reserved.
// Third party copyrights are property of their respective owners.
//
// @Authors
//    Jia Haipeng, jiahaipeng95@gmail.com
//
// Redistribution and use in source and binary forms, with or without modification,
// are permitted provided that the following conditions are met:
//
//   * Redistribution's of source code must retain the above copyright notice,
//     this list of conditions and the following disclaimer.
//
//   * Redistribution's in binary form must reproduce the above copyright notice,
//     this list of conditions and the following disclaimer in the documentation
//     and/or other materials provided with the distribution.
//
//   * The name of the copyright holders may not be used to endorse or promote products
//     derived from this software without specific prior written permission.
//
// This software is provided by the copyright holders and contributors as is and
// any express or implied warranties, including, but not limited to, the implied
// warranties of merchantability and fitness for a particular purpose are disclaimed.
// In no event shall the Intel Corporation or contributors be liable for any direct,
// indirect, incidental, special, exemplary, or consequential damages
// (including, but not limited to, procurement of substitute goods or services;
// loss of use, data, or profits; or business interruption) however caused
// and on any theory of liability, whether in contract, strict liability,
// or tort (including negligence or otherwise) arising in any way out of
// the use of this software, even if advised of the possibility of such damage.
//
//M*/

#ifdef DOUBLE_SUPPORT
#ifdef cl_amd_fp64
#pragma OPENCL EXTENSION cl_amd_fp64:enable
#elif defined (cl_khr_fp64)
#pragma OPENCL EXTENSION cl_khr_fp64:enable
#endif
#endif

///////////////////////////////////////////////////////////////////////////////////////////////////////
///////////////////////////////////////////// flip rows ///////////////////////////////////////////////
///////////////////////////////////////////////////////////////////////////////////////////////////////

__kernel void arithm_flip_rows(__global T * src, int src_step, int src_offset,
                               __global T * dst, int dst_step, int dst_offset,
                               int rows, int cols, int thread_rows, int thread_cols)
{
    int x = get_global_id(0);
    int y = get_global_id(1);

    if (x < cols && y < thread_rows)
    {
        int src_index_0 = mad24(y,            src_step, x + src_offset);
        int src_index_1 = mad24(rows - y - 1, src_step, x + src_offset);

        int dst_index_0 = mad24(y,            dst_step, x + dst_offset);
        int dst_index_1 = mad24(rows - y - 1, dst_step, x + dst_offset);

        T data0 = src[src_index_0], data1 = src[src_index_1];

        dst[dst_index_0] = data1;
        dst[dst_index_1] = data0;
    }
}

__kernel void arithm_flip_rows_cols(__global T * src, int src_step, int src_offset,
                                    __global T * dst, int dst_step, int dst_offset,
                                    int rows, int cols, int thread_rows, int thread_cols)
{
    int x = get_global_id(0);
    int y = get_global_id(1);

    if (x < cols && y < thread_rows)
    {
        int src_index_0 = mad24(y,            src_step, x            + src_offset);
        int dst_index_0 = mad24(rows - y - 1, dst_step, cols - x - 1 + dst_offset);

        int src_index_1 = mad24(rows - y - 1, src_step, cols - x - 1 + src_offset);
        int dst_index_1 = mad24(y,            dst_step, x            + dst_offset);

        T data0 = src[src_index_0], data1 = src[src_index_1];

        dst[dst_index_0] = data0;
        dst[dst_index_1] = data1;
    }
}

///////////////////////////////////////////////////////////////////////////////////////////////////////
///////////////////////////////////////////// flip cols ///////////////////////////////////////////////
///////////////////////////////////////////////////////////////////////////////////////////////////////

__kernel void arithm_flip_cols(__global T * src, int src_step, int src_offset,
                               __global T * dst, int dst_step, int dst_offset,
                               int rows, int cols, int thread_rows, int thread_cols)
{
    int x = get_global_id(0);
    int y = get_global_id(1);

    if (x < thread_cols && y < rows)
    {
        int src_index_0 = mad24(y, src_step, x            + src_offset);
        int dst_index_0 = mad24(y, dst_step, cols - x - 1 + dst_offset);

        int src_index_1 = mad24(y, src_step, cols - x - 1 + src_offset);
        int dst_index_1 = mad24(y, dst_step, x            + dst_offset);

        T data0 = src[src_index_0], data1 = src[src_index_1];
        dst[dst_index_1] = data1;
        dst[dst_index_0] = data0;
    }
}
